NASA Technical Memorandum 87644

NASA-TM-87644 19860013312

A NOVEL METHOD OF CALCULATING FAR-FIELD PATTERNS OF LARGE APERTURE ANTENNAS

M. C. Bailey

MARCH 1986

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665

A NOVEL METHOD OF CALCULATING FAR-FIELD PATTERNS OF LARGE APERTURE ANTENNAS

M. C. Bailey

ABSTRACT

A method is described for calculation of the radiation pattern of large aperture antennas. A piece-wise linear approximation of the aperture field using overlapping pyramidal basis functions allows the radiation pattern of an aperture antenna to be calculated as though it were a two-dimensional array. The calculation of radiation pattern data versus θ and ϕ, suitable for 3-D or contour plot algorithms, is achieved by locating the array in the yz-plane and performing a summation over the aperture field data sampled on a square grid. A FORTRAN subroutine is provided for performing radiation pattern calculations. Numerical results are included to demonstrate the accuracy and convergence of the method. These numerical results indicate that typical accuracies of ±0.1dB for Directivity, ±1dB for the 1st Sidelobe Level, and ±2dB for the 2nd Sidelobe Level can be obtained with an aperture grid of 45×45 points and requires approximately 0.02 seconds CPU time per far-field data point on a VAX 11/750 with a floating point accelerator.

INTRODUCTION

This paper describes a method for calculating the radiation pattern of a large aperture antenna. In order to avoid the necessity of performing a double numerical integration to calculate the radiation patterns, overlapping pyramidal functions are used to develop a piece-wise linear representation of the aperture field. This representation allows the radiation pattern of the aperture antenna to be calculated as though the antenna is a two-dimensional array antenna with an array element pattern described by the radiation pattern of a small square aperture with a pyramidal distribution. The problem of calculating the radiation pattern of the large aperture then becomes one of calculating the array factor of a two-dimensional array and multiplying by the array element pattern. The advantage of this approach is that the double numerical integration becomes a summation over the complex amplitudes of the aperture field at the sampled grid points weighted with an exponential factor corresponding to the grid point location. The problem is formulated such that the array (i.e., aperture) is located in the yz-plane. When the aperture is in the yz-plane, the far-field radiation pattern calculation over $\theta\phi$-space can be accomplished more efficiently.

The present problem is formulated by first expressing the far-field radiation pattern of an aperture located in the yz-plane in terms of the Fourier transforms of the tangential aperture electric fields. A piece-wise linear representation of the aperture field distribution is obtained by the superposition of 3 two-dimensional arrays of pyramids. The three arrays are interleaved such that the superposition yields a 3-dimensional surface approximated by triangular facets. The Fourier transform of this distribution is expressed in a series form which allows an efficient computer code to be written for the radiation pattern calculation. In order to implement this approach, the aperture field must be sampled over a rectangular area with the same sample spacing in both directions. If the aperture boundary is not rectangular, the field values at the sample points outside the aperture must be set to zero.

SYMBOLS

a	Grid spacing for sampled aperture field data.
$A_m(k_y,k_z)$	Series defined in equations (28a) and (29a).
$B_m(k_y,k_z)$	Series defined in equations (28b) and (29b).
$E_a(y,z)$	Electric field distribution in aperture.
$\mathbf{E}_a$	Vector electric field in aperture.
E_y & E_z	y- and z-components of $\mathbf{E}_a$.
E_θ & E_ϕ	θ- and ϕ-components of electric field in far-field region.
E^o_{nm}	Complex amplitude of aperture electric field at sample point (y^o_m,z^o_n).
E^A_{nm} & E^B_{nm}	Definition of type-A and type-B pyramidal basis functions.
$f(k_y,k_z)$	Fourier transform of aperture electric field distribution.
$f^A_o(k_y,k_z)$ $f^B_o(k_y,k_z)$	Fourier transforms of type-A and type-B basis functions.
$f_y(k_y,k_z)$ $f_z(k_y,k_z)$	Fourier transforms of y- and z-components of aperture electric field.
$\mathbf{F}$	Electric vector potential function.
F_θ & F_ϕ	θ- and ϕ-components of $\mathbf{F}$.
j	$=\sqrt{-1}$
k	$=2\pi/\lambda$, Wave number in free space.
k_y & k_z	Fourier transform variables in y- and z-directions.
m	Grid index in y-direction.
M	Total number of sample grid lines in y-direction.
$\mathbf{M}_s$	Equivalent vector magnetic current in aperture.

n	Grid index in z-direction.
$\mathbf{n}$	Unit vector normal to aperture.
N	Total number of sample grid lines in z-direction.
r,θ,ϕ	Coordinate variables in spherical coordinate system.
r_0	Radius of circular aperture.
S_a	Area of aperture.
w	Width of square aperture.
x,y,z	Coordinate variables in cartesian coordinate system.
$\mathbf{y}$ & $\mathbf{z}$	Unit vectors in y- and z-directions.
y_m & z_n	Variables defined in equation (14).
y_{nm} & z_{nm}	Variables defined in equation (15).
y_m^o	y-coordinate of m-th sampling grid line in aperture plane.
z_n^o	z-coordinate of n-th sampling grid line in aperture plane.
λ	Wavelength in free space.

METHOD

The electric field intensity in the far-field region of an antenna is given by

$$E_\theta = -jk\, F_\phi \tag{1}$$

$$E_\phi = jk\, F_\theta \tag{2}$$

where F_θ and F_ϕ are the spherical components of the vector potential function

$$F = \frac{e^{-jkr}}{4\pi r} \iint_{S_a} \mathbf{M}_s\, e^{jk\,(y \sin\theta \sin\phi + z \cos\theta)}\, dy\, dz \tag{3}$$

where the equivalent magnetic current, $\mathbf{M}_s$, is obtained from the tangential electric field, E_a, in the aperture (x=0)

$$\mathbf{M}_s = -2\mathbf{n} \times E_a \tag{4}$$

$$\mathbf{M}_s = 2\,(E_z \mathbf{y} - E_y \mathbf{z}) \tag{5}$$

By substituting

$$k_y = k \sin\theta \sin\phi \tag{6}$$

$$k_z = k \cos\theta \tag{7}$$

the spherical components of the radiated field become

$$E_\theta = \frac{-jk\, e^{-jkr}}{2\pi r} \left[f_z(k_y,k_z) \cos\phi \right] \tag{8}$$

$$E_\phi = \frac{jk\, e^{-jkr}}{2\pi r} \left[f_z(k_y,k_z) \cos\theta \sin\phi + f_y(k_y,k_z) \sin\theta \right] \tag{9}$$

where $f_y(k_y,k_z)$ and $f_z(k_y,k_z)$ are the Fourier transforms of the aperture electric fields defined as

$$f_y(k_y,k_z) = \iint_{S_a} E_y(y,z)\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \tag{10}$$

$$f_z(k_y,k_z) = \iint_{S_a} E_z(y,z)\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \qquad (11)$$

In order to calculate the radiation fields, the integrals in (10) and (11) must be evaluated. For practical aperture field distributions, these integrals have to be evaluated numerically. In the case of sampled aperture data, in which the data are on a rectangular grid, Fast Fourier Transforms can be used to evaluate the far-field. However, use of FFT algorithms results in far-field data which are also spaced on a rectangular grid in the transform variables, k_y and k_z, instead of θ and ϕ as is sometimes desired in order to utilize existing 3-D or contour plotting algorithms.

Another approach would be to expand the aperture field in a set of basis functions whose Fourier transforms can be evaluated in closed form. The basis function used here is a pyramid with a square base, a height equal to the sampled aperture field and with the apex located at the sampled aperture data point. Use of the pyramidal pulse basis function allows the aperture field to be approximated by a piece-wise linear representation in three-dimensions. In order to accomplish this piece-wise linear representation, two types of pyramidal basis functions are needed as defined by

type-A:

$$E^A_{nm} = \begin{cases} E^o_{nm}\,(1 - z_n/a) & \{0 \le z_n \le a\ ;\ -z_n \le y_m \le z_n\} \\ E^o_{nm}\,(1 - y_m/a) & \{0 \le y_m \le a\ ;\ -y_m \le z_n \le y_m\} \\ E^o_{nm}\,(1 + z_n/a) & \{-a \le z_n \le 0\ ;\ z_n \le y_m \le -z_n\} \\ E^o_{nm}\,(1 + y_m/a) & \{-a \le y_m \le 0\ ;\ y_m \le z_n \le -y_m\} \end{cases} \qquad (12)$$

type-B:

$$E^B_{nm} = \begin{cases} E^o_{nm}\,[1 - (y_m+z_n)/a] & \{0 \le y_m \le a\ ;\ 0 \le z_n \le (a-y_m)\} \\ E^o_{nm}\,[1 - (y_m-z_n)/a] & \{0 \le y_m \le a\ ;\ -(a-y_m) \le z_n \le 0\} \\ E^o_{nm}\,[1 + (y_m+z_n)/a] & \{-a \le y_m \le 0\ ;\ -(a+y_m) \le z_n \le 0\} \\ E^o_{nm}\,[1 + (y_m-z_n)/a] & \{-a \le y_m \le 0\ ;\ 0 \le z_n \le (a+y_m)\} \end{cases} \qquad (13)$$

where,

$$y_m = y - y_m^o \tag{14}$$

$$z_n = z - z_n^o \tag{15}$$

The location of the sampled aperture data point is defined by (y_m^o, z_n^o), the amplitude of the sampled aperture field is represented by the complex variable E_{nm}^o, and a represents the sampling interval in the aperture plane.

Figure 1 demonstrates the use of these pyramidal basis functions to approximate a $\cos(\pi y/w)\cos(\pi z/w)$ distribution using 25 (i.e. 5×5) sample points. The piece-wise linear approximation is obtained by the superposition of two interleaved arrays of type-A basis functions with one array of type-B basis functions. This interleaving of arrays of pyramidal basis functions results in an aperture distribution given by

$$E_a(y,z) = \sum_{m_o}\sum_{n_o} E_{nm}^A + \sum_{m_e}\sum_{n_e} E_{nm}^A + \sum_{m_o}\sum_{n_e} E_{nm}^B + \sum_{m_e}\sum_{n_o} E_{nm}^B \tag{16}$$

where m_e and n_e indicate even values of m and n; whereas, m_o and n_o indicate odd values of m and n. The Fourier transform of (16) is given by

$$\begin{aligned} f(k_y,k_z) = \sum_{m_o}\sum_{n_o} E_{nm}^o \Big\{ & \iint (1-z_n/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \\ & + \iint (1-y_m/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \\ & + \iint (1+z_n/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \\ & + \iint (1+y_m/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \Big\} \\ + \sum_{m_e}\sum_{n_e} E_{nm}^o \Big\{ & \iint (1-z_n/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \\ & + \iint (1-y_m/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \\ & + \iint (1+z_n/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \\ & + \iint (1+y_m/a)\, e^{jk_y y} e^{jk_z z}\, dy\, dz \Big\} \end{aligned}$$

$$
\begin{aligned}
&+\sum_{m_o}\sum_{n_e} E^o_{nm} \Big\{ \iint [1-(y_m+z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \\
&\qquad +\iint [1-(y_m-z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \\
&\qquad +\iint [1+(y_m+z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \\
&\qquad +\iint [1+(y_m-z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \Big\} \\
&+\sum_{m_e}\sum_{n_o} E^o_{nm} \Big\{ \iint [1-(y_m+z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \\
&\qquad +\iint [1-(y_m-z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \\
&\qquad +\iint [1+(y_m+z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \\
&\qquad +\iint [1+(y_m-z_n)/a]\, e^{jk_y y}\, e^{jk_z z}\, dy\, dz \Big\}
\end{aligned}
\tag{17}
$$

Using (14) and (15), equation (17) becomes

$$
\begin{aligned}
f(k_y,k_z) = &\sum_{m_o}\sum_{n_o} E^o_{nm}\, f^A_o(k_y,k_z)\, e^{jk_y y^o_m}\, e^{jk_z z^o_n} \\
&+\sum_{m_e}\sum_{n_e} E^o_{nm}\, f^A_o(k_y,k_z)\, e^{jk_y y^o_m}\, e^{jk_z z^o_n} \\
&+\sum_{m_o}\sum_{n_e} E^o_{nm}\, f^B_o(k_y,k_z)\, e^{jk_y y^o_m}\, e^{jk_z z^o_n} \\
&+\sum_{m_e}\sum_{n_o} E^o_{nm}\, f^B_o(k_y,k_z)\, e^{jk_y y^o_m}\, e^{jk_z z^o_n}
\end{aligned}
\tag{18}
$$

where $f_o^A(k_y,k_z)$ and $f_o^B(k_y,k_z)$ are defined as

$$f_o^A(k_y,k_z) = \int_0^a \int_{-z_n}^{z_n} (1-z_n/a)\, e^{jk_y y_m} e^{jk_z z_n}\, dy_m\, dz_n$$

$$+ \int_0^a \int_{-y_m}^{y_m} (1-y_m/a)\, e^{jk_y y_m} e^{jk_z z_n}\, dz_n\, dy_m$$

$$+ \int_{-a}^0 \int_{z_n}^{-z_n} (1+z_n/a)\, e^{jk_y y_m} e^{jk_z z_n}\, dy_m\, dz_n$$

$$+ \int_{-a}^0 \int_{y_m}^{-y_m} (1+y_m/a)\, e^{jk_y y_m} e^{jk_z z_n}\, dz_n\, dy_m \qquad (19)$$

$$f_o^B(k_y,k_z) = \int_0^a \int_0^{(a-y_m)} [\, 1-(y_m+z_n)/a\,]\, e^{jk_y y_m} e^{jk_z z_n}\, dz_n\, dy_m$$

$$+ \int_0^a \int_{-(a-y_m)}^0 [\, 1-(y_m-z_n)/a\,]\, e^{jk_y y_m} e^{jk_z z_n}\, dz_n\, dy_m$$

$$+ \int_{-a}^0 \int_{-(a+y_m)}^0 [\, 1+(y_m+z_n)/a\,]\, e^{jk_y y_m} e^{jk_z z_n}\, dz_n\, dy_m$$

$$+ \int_{-a}^0 \int_0^{(a+y_m)} [\, 1+(y_m-z_n)/a\,]\, e^{jk_y y_m} e^{jk_z z_n}\, dz_n\, dy_m \qquad (20)$$

$f_o^A(k_y,k_z)$ and $f_o^B(k_y,k_z)$ are the Fourier transforms of (12) and (13) with $E_{nm}^o=1$

and $y_m^o = z_n^o = 0$. After much algebraic manipulation, $f_o^A(k_y,k_z)$ and $f_o^B(k_y,k_z)$ can be expressed as

$$f_o^A(k_y,k_z) = \left[\frac{4a^2}{(k_y a)^2 - (k_z a)^2} \right] \left[\frac{\sin(k_y a)\cos(k_z a)}{k_y a} - \frac{\sin(k_z a)\cos(k_y a)}{k_z a} \right] \quad (21)$$

$$f_o^B(k_y,k_z) = \left[\frac{4a^2}{(k_y a)^2 - (k_z a)^2} \right] \left[\frac{\sin(k_z a)}{k_z a} - \frac{\sin(k_y a)}{k_y a} \right] \quad (22)$$

with limiting cases given by

$$f_o^A(k_y,k_y) = 2 \left[\frac{1}{k_y} \right]^2 \left[1 - \sin(2k_y a)/(2k_y a) \right] \quad (23)$$

$$f_o^B(k_y,k_y) = 2 \left[\frac{1}{k_y} \right]^2 \left[\sin(k_y a)/(k_y a) - \cos(k_y a) \right] \quad (24)$$

$$f_o^A(0,0) = \frac{4a^2}{3} \quad (25)$$

$$f_o^B(0,0) = \frac{2a^2}{3} \quad (26)$$

The Fourier transform of the aperture field can now be written as

$$f(k_y,k_z) = \sum_{m=1}^{M} \left[A_m(k_y,k_z)\, f_o^A(k_y,k_z) + B_m(k_y,k_z)\, f_o^B(k_y,k_z) \right] e^{jy_m^o k_y} \quad (27)$$

where, for m=1,3,5,...,M

$$A_m(k_y,k_z) = \sum_{n=1}^{N} E^o_{nm}\, e^{jz^o_n k_z} \qquad \text{(n odd)} \tag{28a}$$

$$B_m(k_y,k_z) = \sum_{n=2}^{N-1} E^o_{nm}\, e^{jz^o_n k_z} \qquad \text{(n even)} \tag{28b}$$

whereas, for m=2,4,6,...,M-1

$$A_m(k_y,k_z) = \sum_{n=2}^{N-1} E^o_{nm}\, e^{jz^o_n k_z} \qquad \text{(n even)} \tag{29a}$$

$$B_m(k_y,k_z) = \sum_{n=1}^{N} E^o_{nm}\, e^{jz^o_n k_z} \qquad \text{(n odd)} \tag{29b}$$

and where both M and N are assumed to be odd.

Using (27) to represent the aperture field transforms in (8) and (9) allows the far-field pattern of the large aperture to be calculated from interleaved arrays, whose element patterns are obtained from (21) and (22).

RESULTS

A FORTRAN subroutine for calculating the far-field pattern of an aperture is given in the Appendix. This subroutine was used to evaluate the accuracy and convergence of the method by calculating the far-field patterns for circular aperture field distributions specified at discrete points on a square grid of N×N grid lines. The distributions used in the evaluation were obtained by sampling the expression $E_y(r)=C+(1-C)[1-(r/r_0)^2]$ over a square grid for values of C equal to 0.0, 0.316 and 1.0, as shown in figure 2. These piece-wise linear aperture distributions are shown in figures 3-5 for grid densities of 15×15, 35×35 and 55×55. Three-dimensional plots of the radiation patterns for an aperture diameter of 1000λ are shown in figures 6-8. These patterns exhibit the characteristic ring sidelobes expected from a circular aperture with a r-dependent excitation distribution; however, careful examination of the patterns for N=15 shows that each of the sidelobes, after the first, has an undulating character as it encircles the main beam. This becomes more obvious for the contour radiation patterns in figures 9-11. Since very little change is noticed between N=35 and N=55, these data tend to indicate that a grid density of 35×35 may be sufficiently dense for accurate radiation pattern calculations; however, one should be cautioned about the possible occurrence of "grating" lobes in the calculated patterns caused by large spacings between aperture sample points. The angular regions where these grating lobes can occur are predictable and should be avoided when calculating radiation patterns.

In order to better establish the minimum grid density needed, the radiation pattern parameters of directivity, beamwidth and the first two sidelobe levels were calculated in the $\theta=90^\circ$ plane over a wide range of grid densities. These data are plotted in figures 12-15. These data indicate that very accurate radiation patterns can be calculated with this method by using an aperture grid density of only 45×45 sample points; however, one must be reminded that this convergence criterion was established for the class of circular aperture distributions defined as a parabola-on-a-pedestal and is intended to only be a guide as to the grid density necessary. One should always check the convergence of the sampling density for the type of distribution one expects for his particular application.

In order to determine the computational efficiency of the method, far-field patterns were calculated over a square grid in $\theta\phi$-coordinates ($81\times81=6561$ far-field points). The average CPU time per far-field point is plotted in figure 16 versus the number of aperture grid lines. The CPU time in figure 16 was obtained on a VAX 11/750 with a floating-point accelerator installed. The corresponding CPU time on an IBM-PC with an 8087 installed is plotted in figure 17. It was observed that the average CPU time per far-field point remained nearly constant as the number of far-field points increased. A note of caution: the CPU time per far-field point will probably be different for a $\theta\phi$ grid which is not square (i.e. the number of θ and ϕ pattern angles are not equal). The CPU times in figures 16 and 17 were each based on a single subroutine call and includes the time required to calculate both the θ- and ϕ-components of the far-field from both the y- and z-components of the aperture field.

CONCLUSION

An efficient numerical method of calculating the volumetric (i.e. over a $\theta\phi$-region) radiation pattern of large aperture antennas has been described. Examples have been given which demonstrate that very accurate results can be expected with an aperture field sample grid of 45×45 points and requires only 0.02 seconds CPU time per far-field point on a VAX 11/750 with a floating-point accelerator installed.



Figure 1. Interleaved arrays of pyramidal basis functions to form piece-wise linear representation of 3-D surface.



Figure 2. Circular aperture field distributions.
$E_y(r) = C + (1-C)[1-(r/r_o)^2]$



Figure 3. Piece-wise linear representation of aperture field. $E_y(r) = 1-(r/r_o)^2$



Figure 4. Piece-wise linear representation of aperture field.
$E_y(r) = 0.316 + (1-0.316)[1-(r/r_o)^2$



Figure 5. Piece-wise linear representation of aperture field. $E_y(r) = 1$



Figure 6. Three-dimensional plots of calculated far-field patterns for aperture distributions of figure 3. ($2r_o=1000\lambda$)



Figure 7. Three-dimensional plots of calculated far-field patterns for aperture distributions of figure 4. ($2r_o=1000\lambda$)



Figure 8. Three-dimensional plots of calculated far-field patterns for aperture distributions of figure 5. ($2r_o=1000\lambda$)



Figure 9. Contour plots of calculated far-field patterns for aperture distributions of figure 3. (5dB increments) ($2r_o$=1000λ)



Figure 10. Contour plots of calculated far-field patterns for aperture distributions of figure 4. (5dB increments) ($2r_o$=1000λ)



Figure 11. Contour plots of calculated far-field patterns for aperture distributions of figure 5. (5dB increments) ($2r_o$=1000λ)



Figure 12. Calculated far-field directivity versus circular aperture grid density (M=N). $E_y(r)=C+(1-C)[1-(r/r_0)^2]$



Figure 13. Calculated far-field 3dB beamwidth versus circular aperture grid density (M=N). $E_y(r)=C+(1-C)[1-(r/r_0)^2]$



Figure 14. Calculated far-field 1st sidelobe level versus circular aperture grid density (M=N).
$E_y(r)=C+(1-C)[1-(r/r_0)^2]$



Figure 15. Calculated far-field 2nd sidelobe level versus circular aperture grid density (M=N).
$E_y(r)=C+(1-C)[1-(r/r_0)^2]$



Figure 16. VAX 11/750 CPU time for 81X81 far-field points versus aperture grid density (M=N).



Figure 17. IBM-PC CPU time for 81X81 far-field points versus aperture grid density (M=N).

APPENDIX

Subroutine RADPAT

This subroutine calculates the radiation pattern over a specified $\theta\phi$ angular region in the far-field of an aperture. The amplitude and phase of the aperture electric field distribution are specified at discrete points over a rectangular area which encloses the aperture boundaries. The discrete field values in the aperture plane are constrained to have the same sample spacing in both directions and the number of samples in both directions must be an odd number. The parameters in the CALL statement to the subroutine are defined by comment statements at the beginning of the subroutine listing.

```
      SUBROUTINE RADPAT(DD,NN,MM,NZ,MY,EYM,EYP,EZM,EZP,AA,
     +                  T1,T2,T3,P1,P2,P3)
C*****************************************************************
C   COMPUTES RADIATION PATTERN FOR APERTURE WITH DISTRIBUTION   *
C   SPECIFIED AS POINTS ON SQUARE GRID IN YZ-PLANE.             *
C                                                               *
C     METHOD AND PROGRAM BY:   M. C. BAILEY                     *
C                                                               *
C     THETA, PHI, DBT AND DBP OUTPUT TO UNIT # 12               *
C                                                               *
C     STEP ANGLE = THETA                                        *
C     SCAN ANGLE = PHI                                          *
C     BORESITE DIRECTION IS (THETA=90,PHI=0)                    *
C*****************************************************************
C     DD = GRID SPACING IN WAVELENGTHS. (SQUARE GRID)           *
C     NN = NUMBER (ODD) OF APERTURE FIELD POINTS IN Z-DIRECTION*
C     MM = NUMBER (ODD) OF APERTURE FIELD POINTS IN Y-DIRECTION*
C     NZ,MY = ARRAY DIMENSIONS (SEE NOTE BELOW).                *
C                                                               *
C     APERTURE ELECTRIC FIELD AT (Z-SUBN,Y-SUBM):               *
C          EYM(N,M) = Y-COMPONENT MAGNITUDE                     *
C          EYP(N,M) = Y-COMPONENT PHASE (RADIANS)               *
C          EZM(N,M) = Z-COMPONENT MAGNITUDE                     *
C          EZP(N,M) = Z-COMPONENT PHASE (RADIANS)               *
C                                                               *
C     AA = TEMPORARY STORAGE ARRAY.                             *
C                                                               *
C     PATTERN ANGLES (DEGREES):                                 *
C          T1,T2,T3 = THETA (INITIAL,FINAL,INCREMENT)           *
C          P1,P2,P3 = PHI   (INITIAL,FINAL,INCREMENT)           *
C                                                               *
C      NOTE:  IN CALLING PROGRAM;                               *
C             EYM, EYP, EZM, EZP, AA                            *
C             MUST BE DIMENSIONED AS FOLLOWS:                   *
C                                                               *
C             EYM(NZ,MY),EYP(NZ,MY)                             *
C             EZM(NZ,MY),EZP(NZ,MY)                             *
C             AA(8,MY)                                          *
C*****************************************************************
      DIMENSION EYM(NZ,MY),EYP(NZ,MY)
      DIMENSION EZM(NZ,MY),EZP(NZ,MY)
      DIMENSION AA(8,MY)
      PI=2.*ASIN(1.0)
      PIO2=0.5*PI
      DTOR=PI/180.
      AK=2.*PI*DD
      MM1=MM-1
      NN1=NN-1
```

```
      IF((MM1-2*(MM1/2)).EQ.0) GO TO 20
      MM=MM+1
      MM1=MM-1
      IF(MM.GT.MY) GO TO 300
      DO 10 N=1,NN
      EYM(N,MM)=0.0
      EYP(N,MM)=0.0
      EZM(N,MM)=0.0
      EZP(N,MM)=0.0
10    CONTINUE
20    IF((NN1-2*(NN1/2)).EQ.0) GO TO 40
      NN=NN+1
      NN1=NN-1
      IF(NN.GT.NZ) GO TO 300
      DO 30 M=1,MM
      EYM(NN,M)=0.0
      EYP(NN,M)=0.0
      EZM(NN,M)=0.0
      EZP(NN,M)=0.0
30    CONTINUE
40    NT=(NN+1)/2
      MT=(MM+1)/2
      DDSQ=DD*DD
      C0=4.0*DDSQ
      C2=2.0*DDSQ/3.0
      C1=2.0*C2
      U0=1.0
      PA=0.0
      PB=0.0
      DO 70 N=1,NN,2
      DO 50 M=1,MM,2
      PA=PA+(EYM(N,M)*EYM(N,M)+EZM(N,M)*EZM(N,M))
50    CONTINUE
      DO 60 M=2,MM1,2
      PB=PB+(EYM(N,M)*EYM(N,M)+EZM(N,M)*EZM(N,M))
60    CONTINUE
70    CONTINUE
      DO 100 N=2,NN1,2
      DO 80 M=2,MM1,2
      PA=PA+(EYM(N,M)*EYM(N,M)+EZM(N,M)*EZM(N,M))
80    CONTINUE
      DO 90 M=1,MM,2
      PB=PB+(EYM(N,M)*EYM(N,M)+EZM(N,M)*EZM(N,M))
90    CONTINUE
100   CONTINUE
      PRD=(C1*PA+C2*PB)
110   U0=PRD/(4.*PI)
```

```
      DO 120 M=1,MM
      DO 120 N=1,NN
      A=EYM(N,M)
      P=EYP(N,M)
      EYM(N,M)=A*COS(P)
      EYP(N,M)=A*SIN(P)
      A=EZM(N,M)
      P=EZP(N,M)
      EZM(N,M)=A*COS(P)
      EZP(N,M)=A*SIN(P)
120   CONTINUE
      NUMTH=(T2-T1)/T3+1.5
      NUMPH=(P2-P1)/P3+1.5
      WRITE(12,130)NUMTH,NUMPH
130   FORMAT(2I5)
      DO 290 I=1,NUMTH
      THETA=T1+(I-1)*T3
      TH=THETA*DTOR
      CT=COS(TH)
      ST=SIN(TH)
      AKZ=AK*CT
      AKS=AK*ST
      SAZ=1.0
      IF(ABS(AKZ).GT.1.E-05)SAZ=SIN(AKZ)/AKZ
      CAKZ=COS(AKZ)
      AKZSQ=AKZ*AKZ
      DO 140 M=1,MM
      DO 140 N=1,8
      AA(N,M)=0.0
140   CONTINUE
      DO 170 N=1,NN,2
      EN=(N-NT)*AKZ
      ER=COS(EN)
      EI=SIN(EN)
      DO 150 M=1,MM,2
      AA(1,M)=AA(1,M)+(ER*EYM(N,M)-EI*EYP(N,M))
      AA(2,M)=AA(2,M)+(EI*EYM(N,M)+ER*EYP(N,M))
      AA(3,M)=AA(3,M)+(ER*EZM(N,M)-EI*EZP(N,M))
      AA(4,M)=AA(4,M)+(EI*EZM(N,M)+ER*EZP(N,M))
150   CONTINUE
      DO 160 M=2,MM1,2
      AA(5,M)=AA(5,M)+(ER*EYM(N,M)-EI*EYP(N,M))
      AA(6,M)=AA(6,M)+(EI*EYM(N,M)+ER*EYP(N,M))
      AA(7,M)=AA(7,M)+(ER*EZM(N,M)-EI*EZP(N,M))
      AA(8,M)=AA(8,M)+(EI*EZM(N,M)+ER*EZP(N,M))
160   CONTINUE
170   CONTINUE
```

```
      DO 200 N=2,NN1,2
      EN=(N-NT)*AKZ
      ER=COS(EN)
      EI=SIN(EN)
      DO 180 M=2,MM1,2
      AA(1,M)=AA(1,M)+(ER*EYM(N,M)-EI*EYP(N,M))
      AA(2,M)=AA(2,M)+(EI*EYM(N,M)+ER*EYP(N,M))
      AA(3,M)=AA(3,M)+(ER*EZM(N,M)-EI*EZP(N,M))
      AA(4,M)=AA(4,M)+(EI*EZM(N,M)+ER*EZP(N,M))
180   CONTINUE
      DO 190 M=1,MM,2
      AA(5,M)=AA(5,M)+(ER*EYM(N,M)-EI*EYP(N,M))
      AA(6,M)=AA(6,M)+(EI*EYM(N,M)+ER*EYP(N,M))
      AA(7,M)=AA(7,M)+(ER*EZM(N,M)-EI*EZP(N,M))
      AA(8,M)=AA(8,M)+(EI*EZM(N,M)+ER*EZP(N,M))
190   CONTINUE
200   CONTINUE
210   DO 280 J=1,NUMPH
      PHI=P1+(J-1)*P3
      PH=PHI*DTOR
      SP=SIN(PH)
      CP=COS(PH)
      AKY=AKS*SP
      AKYSQ=AKY*AKY
      AKYZSQ=AKYSQ-AKZSQ
      IF(ABS(AKYZSQ).LT.1.E-05) GO TO 220
      CC=C0/AKYZSQ
      SAY=1.0
      IF(ABS(AKY).GT.1.E-05)SAY=SIN(AKY)/AKY
      CAKY=COS(AKY)
      FA=CC*(CAKZ*SAY-CAKY*SAZ)
      FB=CC*(SAZ-SAY)
      GO TO 240
220   IF(ABS(AKY).LT.0.005) GO TO 230
      IF(ABS(AKZ).LT.0.005) GO TO 230
      CC=-2.0*DDSQ/(AKY**3)
      SAKY=SIN(AKY)
      CAKY=COS(AKY)
      FA=CC*(SAKY*CAKY-AKY)
      FB=CC*(AKY*CAKY-SAKY)
      GO TO 240
230   FA=C1
      FB=C2
240   CONTINUE
      FYR=0.0
      FYI=0.0
      FZR=0.0
      FZI=0.0
```

```
      DO 250 M=1,MM
      EM=(M-MT)*AKY
      ER=COS(EM)
      EI=SIN(EM)
      FYR=FYR+(FA*(ER*AA(1,M)-EI*AA(2,M))
     &         +FB*(ER*AA(5,M)-EI*AA(6,M)))
      FYI=FYI+(FA*(EI*AA(1,M)+ER*AA(2,M))
     &         +FB*(EI*AA(5,M)+ER*AA(6,M)))
      FZR=FZR+(FA*(ER*AA(3,M)-EI*AA(4,M))
     &         +FB*(ER*AA(7,M)-EI*AA(8,M)))
      FZI=FZI+(FA*(EI*AA(3,M)+ER*AA(4,M))
     &         +FB*(EI*AA(7,M)+ER*AA(8,M)))
250   CONTINUE
      ETR=FZR*CP
      ETI=FZI*CP
      EPR=FYR*ST+FZR*CT*SP
      EPI=FYI*ST+FZI*CT*SP
      UTH=ETR*ETR+ETI*ETI
      UPH=EPR*EPR+EPI*EPI
      GTH=UTH/U0
      GPH=UPH/U0
260   DBT=-150.0
      DBP=-150.0
      IF(GTH.GT.1.E-15)DBT=10.*ALOG10(GTH)
      IF(GPH.GT.1.E-15)DBP=10.*ALOG10(GPH)
      WRITE(12,270)THETA,PHI,DBT,DBP
270   FORMAT(4F10.4)
280   CONTINUE
290   CONTINUE
      RETURN
300   WRITE(12,310)
310   FORMAT(1X42HNUMBER OF SAMPLE POINTS EXCEEDS ARRAY SIZE/)
      STOP
      END
```

Standard Bibliographic Page

1. Report No. NASA TM-87644	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle A Novel Method of Calculating Far-Field Patterns of Large Aperture Antennas		5. Report Date March 1986 6. Performing Organization Code 506-44-21-03
7. Author(s) M. C. Bailey		8. Performing Organization Report No.
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665-5225		10. Work Unit No. 11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		13. Type of Report and Period Covered Technical Memorandum 14. Sponsoring Agency Code
15. Supplementary Notes		

16. Abstract

A method is described for calculation of the radiation pattern of large aperture antennas. A piece-wise linear approximation of the aperture field using overlapping pyramidal basis functions allows the radiation pattern of an aperture antenna to be calculated as though it were a two-dimensional array. The calculation of radiation pattern data versus θ and ϕ, suitable for 3-D or contour plot algorithms, is achieved by locating the array in the yz-plane and performing a summation over the aperture field data sampled on a square grid. A FORTRAN subroutine is provided for performing radiation pattern calculations. Numerical results are included to demonstrate the accuracy and convergence of the method. These numerical results indicate that typical accuracies of ±0.1dB for Directivity, ±1dB for the 1st Sidelobe Level, and ±2dB for the 2nd Sidelobe Level can be obtained with an aperture grid of 45×45 points and requires approximately 0.02 seconds CPU time per far-field data point on a VAX 11/750 with a floating point accelerator.

17. Key Words (Suggested by Authors(s))	18. Distribution Statement
Antennas Apertures Reflectors Radiation Patterns	Unclassified - Unlimited Subject Category 32

19. Security Classif.(of this report)	20. Security Classif.(of this page)	21. No. of Pages	22. Price
Unclassified	Unclassified	38	A03

For sale by the National Technical Information Service, Springfield, Virginia 22161